Здравствуйте, уже несколько часов залипаю с этим вопросом.. Мне в печатную форму надо получить преобразованный Юр. Адрес. В общем модуле нашел эту функцию, но возвращает "Соответствие".

Пишу так :

```1c
    АдресОрг = УправлениеКонтактнойИнформациейЗарплатаКадры.АдресаОрганизаций(ВыборкаОрганизаций.Организация);
```

Вот код с общего модуля, даже с комментариями, но все равно не могу понять, как вытащить Юр. Адрес... Надеюсь на вашу помощь!

```1c
    // Функция возвращает юридический и фактический адреса, указанной организации.
    //
    // Параметры:
    // Организации - Массив ссылок или ссылка на элемент справочника Организации.
    //
    // Возвращаемое значение:
    // Соответствие:
    // Ключ - СправочникСсылка.Организации
    // Значение - Соответствие
    // Ключ - СправочникСсылка.ВидыКонтактнойИнформации
    // Значение - Структура
    // Представление
    // Город
    // ЗначенияПолей
    //
    Функция АдресаОрганизаций(Организации) Экспорт

    ВозвращаемоеЗначение = Новый Соответствие;

    ТипСправочникСсылкаОрганизации = Тип("СправочникСсылка.Организации");

    // Определение соответствия видов контактной информации в зависимости от
    // типа объекта, содержащего контактную информацию.
    СоответствиеАдресовОрганизаций = Новый Соответствие;

    СоответствиеВидов = Новый Соответствие;
    СоответствиеВидов.Вставить(Справочники.ВидыКонтактнойИнформации.ФактАдресОрганизации, Справочники.ВидыКонтактнойИнформации.ФактАдресОрганизации);
    СоответствиеВидов.Вставить(Справочники.ВидыКонтактнойИнформации.ЮрАдресОрганизации, Справочники.ВидыКонтактнойИнформации.ЮрАдресОрганизации);

    СоответствиеАдресовОрганизаций.Вставить(ТипСправочникСсылкаОрганизации, СоответствиеВидов);

    ЗарплатаКадрыПереопределяемый.ДополнитьСоответствиеАдресовОрганизаций(СоответствиеАдресовОрганизаций);

    // Деление организаций по типу объекта, содержащего контактную информацию.
    КоллекцияПоТипам = Новый Соответствие;

    Если ТипЗнч(Организации) = ТипСправочникСсылкаОрганизации Тогда
    КоллекцияПоТипам.Вставить(ТипСправочникСсылкаОрганизации, ОбщегоНазначенияКлиентСервер.ЗначениеВМассиве(Организации));
    Иначе
    КоллекцияПоТипам.Вставить(ТипСправочникСсылкаОрганизации, Организации);
    КонецЕсли;

    ЗарплатаКадрыПереопределяемый.ОпределитьТипыВладельцевАдресовОрганизаций(КоллекцияПоТипам);

    // Получение адресов
    Для каждого КоллекцияПоТипу Из КоллекцияПоТипам Цикл

    Если КоллекцияПоТипу.Ключ = Тип("СправочникСсылка.Организации") Тогда
    МассивСсылок = КоллекцияПоТипу.Значение;
    СоответствиеВидовКИ = СоответствиеАдресовОрганизаций.Получить(ТипСправочникСсылкаОрганизации);
    Иначе
    МассивСсылок = ОбщегоНазначения.ВыгрузитьКолонку(КоллекцияПоТипу.Значение, "Ключ");
    СоответствиеВидовКИ = СоответствиеАдресовОрганизаций.Получить(КоллекцияПоТипу.Ключ);
    КонецЕсли;

    ВидыАдресов = ОбщегоНазначения.ВыгрузитьКолонку(СоответствиеВидовКИ, "Ключ");

    Запрос = Новый Запрос;
    Запрос.МенеджерВременныхТаблиц = Новый МенеджерВременныхТаблиц;

    УправлениеКонтактнойИнформацией.СоздатьВТКонтактнаяИнформация(Запрос.МенеджерВременныхТаблиц, МассивСсылок, , ВидыАдресов);

    Запрос.Текст =
    "ВЫБРАТЬ
    | КонтактнаяИнформация.Объект КАК Объект,
    | КонтактнаяИнформация.Вид,
    | КонтактнаяИнформация.Представление,
    | КонтактнаяИнформация.ЗначенияПолей
    |ИЗ
    | ВТКонтактнаяИнформация КАК КонтактнаяИнформация
    |ИТОГИ ПО
    | Объект";

    ВыборкаОрганизаций = Запрос.Выполнить().Выбрать(ОбходРезультатаЗапроса.ПоГруппировкам);
    Пока ВыборкаОрганизаций.Следующий() Цикл

    СоответствиеАдресовОрганизации = Новый Соответствие;
    ВыборкаПоВидам = ВыборкаОрганизаций.Выбрать();
    Пока ВыборкаПоВидам.Следующий() Цикл

    СтруктураАдреса = СтруктураПустогоАдресаОрганизации();
    ЗаполнитьЗначенияСвойств(СтруктураАдреса, ВыборкаПоВидам);

    АдресСтруктура = ЗарплатаКадры.СтруктураАдресаИзXML(
    ВыборкаПоВидам.ЗначенияПолей, ВыборкаПоВидам.Вид);

    Сокращение = ";
    Если АдресСтруктура.Свойство("Город") И НЕ ПустаяСтрока(АдресСтруктура.Город) Тогда
    СтруктураАдреса.Город = АдресСтруктура.Город;
    АдресСтруктура.Свойство("ГородСокращение", Сокращение);
    ИначеЕсли АдресСтруктура.Свойство("НаселенныйПункт") И НЕ ПустаяСтрока(АдресСтруктура.НаселенныйПункт) Тогда
    СтруктураАдреса.Город = АдресСтруктура.НаселенныйПункт;
    АдресСтруктура.Свойство("НаселенныйПунктСокращение", Сокращение);
    ИначеЕсли АдресСтруктура.Свойство("Регион") И НЕ ПустаяСтрока(АдресСтруктура.Регион) Тогда
    СтруктураАдреса.Город = АдресСтруктура.Регион;
    АдресСтруктура.Свойство("РегионСокращение", Сокращение);
    КонецЕсли;

    Если НЕ ПустаяСтрока(СтруктураАдреса.Город) И НЕ ПустаяСтрока(Сокращение) Тогда
    СтруктураАдреса.Город = Сокращение + ". " + Лев(СтруктураАдреса.Город, СтрДлина(СтруктураАдреса.Город) - СтрДлина(Сокращение) - 1);
    КонецЕсли;

    ВидАдреса = СоответствиеВидовКИ.Получить(ВыборкаПоВидам.Вид);
    СоответствиеАдресовОрганизации.Вставить(ВидАдреса, СтруктураАдреса);

    КонецЦикла;

    Если КоллекцияПоТипу.Ключ = Тип("СправочникСсылка.Организации") Тогда
    СсылкаНаОрганизацию = ВыборкаОрганизаций.Объект;
    Иначе
    СсылкаНаОрганизацию = КоллекцияПоТипу.Значение.Получить(ВыборкаОрганизаций.Объект);
    КонецЕсли;
    ВозвращаемоеЗначение.Вставить(СсылкаНаОрганизацию, СоответствиеАдресовОрганизации);

    КонецЦикла;

    КонецЦикла;

    Возврат ВозвращаемоеЗначение;

    КонецФункции
```
---
```1c
    АдресОрг = УправлениеКонтактнойИнформациейЗарплатаКадры.АдресаОрганизаций(ВыборкаОрганизаций.Организация);
    ОписаниеЮридическогоАдреса = УправлениеКонтактнойИнформациейЗарплатаКадры.АдресОрганизации(
    АдресОрг,
    ВыборкаОрганизаций.Организация,
    Справочники.ВидыКонтактнойИнформации.ЮрАдресОрганизации);
    АдресЮридический = ОписаниеЮридическогоАдреса.Представление;
```
---
```1c
    АдресОрг = УправлениеКонтактнойИнформациейЗарплатаКадры.АдресаОрганизаций(ВыборкаОрганизаций.Организация);
    ОписаниеЮридическогоАдреса = УправлениеКонтактнойИнформациейЗарплатаКадры.АдресОрганизации(
    АдресОрг,
    ВыборкаОрганизаций.Организация,
    Справочники.ВидыКонтактнойИнформации.ЮрАдресОрганизации);
    АдресЮридический = ОписаниеЮридическогоАдреса.Представление;
```
---
Cпасибо большое!